

Ch. 07 二維影像的繪圖

影像(image)是用二維陣列來表示，主要分成

色階(indexed)影像 和
全彩(true color)影像

影像種類

種類	說明
二元影像(Binary Images)	可允許2個像素(pixel)值，其值為0或1
灰階影像(Gray Scale Images)	可允許256個像素值，其值為0至255，代表不同的灰階
色階影像(Indexed Images)	可允許256個像素值，其值為0至255，代表不同的顏色
全彩影像(true color image)	RGB Images，R為紅色，G為綠色，B為藍色，每個原色可允許256個像素值，總共可允許 $2^8*2^8*2^8$ (一千6百萬)個顏色

全彩影像的顏色交織(color interleave)

格式	影像變數維度
Pixel Interleave	(3,m,n)，交織維度3在第1維，以點為單位做顏色混合 每一pixel含(r,g,b)各一個值
Line or Row Interleave	(m,3,n)，交織維度3在第2維，以線為單位做顏色混合 每一掃描線，分別以r,g,b三條掃描線組成
Band Interleave	(m,n,3)，交織維度3在第3維，以面為單位做顏色混合 每一影像，分別以r,g,b三張影像所組成

每一掃描線含有多個pixel，每張影像含有多條掃描線。

IDL支援的影像格式

影像格式	全名
BMP	BitMap
GIF	Graphics Interchange Format 檔案小，影像品質尚可，常用在網頁
JPEG或JPG	Joint Photographic Experts Group 為全彩影像，影像清晰且檔案大小適中
PNG	Portable Network Graphics
PPM	Portable PixMap
SRF	Sun RasterFile
TIFF或TIF	Tagged Image File Format 影像格式的品質最好，但檔案大

衛星影像上常用的HDF(Hierarchical Data Format), NetCDF

醫學影像上常用的DICOM(Digital Imaging and Communication in Medicine)

FILEPATH函數

函數	說明
Result = FILEPATH(Filename)	回傳IDL系統檔案的路徑，其中Filename是系統檔案的名稱

FILEPATH函數的關鍵字

關鍵字	說明
SUBDIRECTORY	設定預設系統檔案目錄的子目錄

```
path = filepath('rose.jpg')  
print,path  
subdir=['examples','data']  
path1=FILEPATH('rose.jpg',SUBDIRECTORY=subdir)  
print,path1  
end
```

READ_IMAGE函數語法

語法	說明
Result = READ_IMAGE(Filename[,R,G,B])	輸入檔名Filename，RGB是三原色

WRITE_IMAGE函數語法

語法	說明
WRITE_IMAGE(Filename, Format, Data[,R,G,B])	儲存檔名Filename，影像格式Format，IDL變數Data，RGB是三原色

Format

A scalar string containing the name of the file format to write. The following are the supported formats:

BMP、GIF、JPEG, PNG, PPM, SRF, TIFF

QUERY_IMAGE函數語法

語法	說明
Result = QUERY_IMAGE(Filename)	輸入檔名Filename，查詣影像檔的資訊

QUERY_IMAGE函數的關鍵字

關鍵字	說明
CHANNELS= variable	回影像的頻道數
DIMENSIONS = variable	回傳影像的維度
NUM_IMAGES = variable	回影像的張數
PIXEL_TYPE = variable	回傳像素的資料型態
TYPE = string	回傳影像的格式

查詢影像詳細資訊，以減少影像讀取時的錯誤。


```
info=QUERY_IMAGE('rose.jpg')
```

```
print,info ;回傳0表示影像檔不在目前的工作目錄中
```

```
subdir = ['examples','data']
```

```
path = FILEPATH('rose.jpg',SUBDIRECTORY=subdir)
```

```
info1=QUERY_IMAGE(path)
```

```
print,info1 ;回傳1表示影像檔存在IDL系統目錄中
```

```
info2 = QUERY_IMAGE('ASAS201710311200.JPG')
```

```
print,info2 ;回傳1表示影像檔在目前的工作目錄中
```

```
end
```

```
subdir = ['examples','data']
```

```
path = FILEPATH('rose.jpg',SUBDIRECTORY=subdir)
```

```
info1=QUERY_IMAGE(path, CHANNELS=channels_1,$
```

```
  DIMENSIONS=dims, TYPE=type_1,$
```

```
  PIXEL_TYPE=ptype, NUM_IMAGES=num)
```

```
print,info1 ;回傳1表示影像檔存在IDL系統目錄中
```

```
print,'channels_1=',channels_1
```

```
print,'type_1=', type_1
```

```
print,'dims=',dims
```

```
print,'ptype=',ptype,' num=',num
```

```
info2 = QUERY_IMAGE('ASAS201710311200.JPG', $
```

```
  CHANNELS=channels_1,$
```

```
  DIMENSIONS=dims, TYPE=type_1,$
```

```
  PIXEL_TYPE=ptype, NUM_IMAGES=num)
```

```
print,info2 ;回傳1表示影像檔在目前的工作目錄中
```

```
print,'channels_1=',channels_1
```

```
print,'type_1=', type_1
```

```
print,'dims=',dims
```

```
print,'ptype=',ptype,' num=',num
```

```
end
```

TV程序的語法

語法	說明
TV,Image[,Position]	宣告影像區域號碼Position。 0左上區域, 1右上, 2左下, 3右下
TV,Image(,x,y)	宣告影像左下角位置(X,Y), 影像從視窗的 左下角原點平移到(X,Y)位置

影像像素(pixel)值的範圍是[0,255]時，使用TV程序。
像素值不在此範圍時，則使用TVSCL程序，此程序會自動
把像素值轉換至[0,255]後再顯像。

TV程序的關鍵字

關鍵字	說明
/ORDER	改變影像的顯示順序(display order)，由上位下
TRUE=[1 2 3]	宣告影像的顏色交織維度，1：在第一維度，2：在第二維度，3：在第三維度
XSIZE=value	宣告影像的寬度
YSIZE=value	宣告影像的高度
/CENTIMETERS	宣告長度的單位為公分
/INCHS	宣告長度的單位為吋(英寸)

```
window,xsize=600,ysize=500
```

```
subdir=['examples','data']
```

```
path = FILEPATH('rose.jpg',SUBDIRECTORY=subdir)
```

```
image1 = READ_IMAGE(path)
```

```
help,image1
```

```
path2 = FILEPATH('examples.tif',SUBDIRECTORY=subdir)
```

```
image2 = READ_IMAGE(path2)
```

```
help,image2
```

```
image3 = READ_IMAGE(path2,r,g,b)
```

```
help,image3,r,g,b
```

```
TV,image1,TRUE=1,0
```

```
TV,image1,TRUE=1,3,/ORDER
```

```
TV,image1,TRUE=1
```

```
TV,image1,TRUE=1,300,0
```

```
redChannel=REFORM(image1(0,*,*))
```

```
greenChannel=reform(image1(1,*,*))
```

```
blueChannel=reform(image1(2,*,*))
```

```
help,redChannel
```

```
TV,redChannel,2
```

```
temp2 = CONGRID(redChannel,108,81)
```

```
TV,temp2,300,400
```

```
erase
```

```
imagesize=size(redChannel)
```

```
;TV,redChannel,0
```

```
TV,redChannel
```

```
TV,greenChannel,imagesize[1]/2,imagesize[2]/2
```

```
TV,blueChannel,imagesize[1],imagesize[2]
```

```
end
```

Resize array

- Result = CONGRID(Array, X, Y, Z [, /CENTER] [, CUBIC=value{-1 to 0}] [, /INTERP] [, /MINUS_ONE])

關鍵字	說明
CENTER	假定輸入和輸出數組中的點位於其坐標的中點而不是其左下角。
CUBIC	值在-1和0之間，以使用三次卷積插值方法（指定值作為插值參數）。
INTERP	在調整一維或二維數組大小時，強制CONGRID使用線性插值。
MINUS_ONE	設置此關鍵字可防止CONGRID推斷超出輸入數組範圍的一行或一列。

WRITE_IMAGE, Filename, Format, Data [, Red, Green, Blue] [, /APPEND]

- WRITE_IMAGE過程將圖像及其顏色表矢量（如果有）寫入指定類型的檔案。
- Format
- A scalar string containing the name of the file format to write. The following are the supported formats:

BMP, GIF, JPEG, PNG, PPM, SRF, TIFF

```
window,xsize=600,ysize=500
subdir=['examples','data']
path = FILEPATH('rose.jpg',SUBDIRECTORY=subdir)
image1 = READ_IMAGE(path)
tv,image1,true=1,0
help,image1
image_2=congrid(image1,3,114,75)
;resize image1 to col=114, row=75
help,image_2
tv,image_2,true=1,250,350
write_image,'rose_small.jpg','jpeg',image_2
image_3=congrid(image1,3,450,290); resize image1 to col=450,
row=290
tv,image_3,true=1
write_image,'rose_large.jpg','jpeg',image_3
loadct, 13
write_jpeg,'rose_congrid_1.jpg', TVRD(/TRUE), /TRUE
end
```




```
subdir = ['examples','data']
path = FILEPATH('moon_landing.png',SUBDIRECTORY=subdir)
info1=QUERY_IMAGE(path, CHANNELS=channels_1,$
    DIMENSIONS=dims, TYPE=type_1,$
    PIXEL_TYPE=ptype, NUM_IMAGES=num)
print,info1 ;回傳1表示影像檔存在IDL系統目錄中
print,'channels_1=',channels_1
print,'type_1=', type_1
print,'dims=',dims
print,'ptype=',ptype,' num=',num
image1=read_image(path)
help,image1
TV,image1,0
path2=filepath('Day.jpg',subdirectory=subdir)
info2=query_image(path2,Channels=chan_2,
dimensions=dims_2,type=type_2,pixel_type=ptype_2,$
    num_images=num_2)
print,'channels_2=',chan_2
print,'dims_2=',dims_2
image2=read_image(path2)
TV,image2,true=1,1
```

Window,1

path3=filepath('Night.jpg',subdirectory=subdir)

info3=query_image(path3,Channels=chan_3,Dimensions=dims_3,type=type_3,pixel_type=ptype_3,
\$

num_images=num_3)

print,'chan_3=',chan_3

print,'dims_3=',dims_3

image3=read_image(path3)

TV,image3,true=1

end

BYTSCL函數的語法

語法	功能
Result = BYTSCL(Array)	轉換資料數值至[0,255]，引數Array是輸入影像

當影像變數內的數值不在[0,255]間時，TV程序顯現的影像顏色不對，IDL會解讀超過[0,255]數值的輪迴。

所以需要BYTSCL函數，來轉換像素值至[0,255]之間。

BYTSCL函數的關鍵字

語法	功能
MAX=integer	定義舊資料轉換的最大值
MIN=integer	定義舊資料轉換的最小值
/NAN	忽略資料中無法定義的數
TOP=integer	定義新資料尺度的最大值

```
window,0, xsize=375,ysize=150
subdir=['examples','data']
path2 = FILEPATH('examples.tif',SUBDIRECTORY=subdir)
image2 = READ_IMAGE(path2,r,g,b)
info=QUERY_IMAGE(path2,DIMENSIONS=dims,TYPE=type1,PIXEL_TYPE=ptype,NUM_I
MAGES=num)
print,info,type1,dims
print,ptype,num
TV,image2 ;影像顏色不對,
print,max(image2),min(image2) ;顏色偏下層顏色，黑白系列會偏向黑色
image3=BYTSCL(image2)
print,max(image3),min(image3)
TV,image3
image4=REVERSE(image3,2)
TV,image4

window,1, xsize=375,ysize=150
image5=image2[125:364,0:149]
image6=BYTSCL(image5)
TV,image6
image7=REVERSE(image6,2)
TV,image7
image8=BYTSCL(image5, TOP=150, MAX=30, MIN=10)
;將[10,30]轉換為[0,150]
print,max(image8),min(image8)
window,2, xsize=375,ysize=150
TV,image8
end
```

TVSCL程序

語法	說明
TVSCL, Image[,Position]	宣告影像區域號碼Position，引數 image是輸入影像
TVSCL,Image[,X,Y]	宣告影像左下角位置(X,Y)

TVSCL程序是TV程序和BYTSCL函數的合成，會先以BYTSCL函數轉換像素值範圍至[0,255]，最後再以TV程序顯示像。

TVSCL程序的關鍵字

關鍵字	說明
/ORDER	改變影像顯示的順序，由上往下
TRUE=[1 2 3]	宣告影像的顏色交織維度，1：在第一維，2：在第二維，3：在第三維
XSIZE=value	定義影像的寬度
YSIZE=value	定義影像的高度
/CENTIMETERS	定義長度的單位為公分
/INCHS	定義長度的單位為吋(英寸)
TOP=integer	定義新資料尺度的最大值

```
window,0, xsize=375,ysize=150
subdir=['examples','data']
path = FILEPATH('examples.tif',SUBDIRECTORY=subdir)
image1 = READ_IMAGE(path,r,g,b)
TVSCL,image1
;image2=REVERSE(image1,2)
;TVSCL,image2
TVSCL,image1,/ORDER

end
```

將圖繪製到IDL.ps檔案中

```
subdir=['examples','data']  
path = FILEPATH('examples.tif',SUBDIRECTORY=subdir)  
image1 = READ_IMAGE(path,r,g,b)  
image2=REVERSE(image1,2)  
SET_PLOT,'PS' ;設定繪圖裝置為PS  
DEVICE,FILENAME='new_IDL.ps' ;設定輸出的檔案名稱，若無則內定為idl.ps  
TVSCL,image2,XSIZE=6,YSIZE=4,/INCHES ;將圖繪製到idl.ps檔案中  
DEVICE,/CLOSE_FILE ;關閉檔案  
SET_PLOT,'WIN' ;設定繪圖輸出裝置為Windows視窗  
TVSCL,image2  
end
```


練習3

從媒體櫃中找一圖檔(如Penguins.jpg)，並複製到你目前的IDL工作目錄中。

試寫一IDL程式，(1)查詢並顯示所有關於Penguins.jpg影像的訊息。(2)讀取並顯示在windows視窗。(3)將Penguins.jpg長和高都縮小一半，再將縮小後的影像顯示在windows視窗。(4)將縮小後的影像以相同的格式寫入到一個新檔案中。